2.11 Finite Legendre Series

A. Purpose

This subroutine computes the value of a finite sum of Legendre polynomials,

$$y = \sum_{j=0}^{\mathrm{N}} a_j P_j(x)$$

for a specified summation limit, N, argument, x, and sequence of coefficients, a_j. The Legendre polynomials are defined in [1].

B. Usage

B.1 Program Prototype, Single Precision

INTEGER N

REAL X, Y, A$(0 : m \geq \mathrm{N})$

Assign values to X, N, and A(0), A(1), ... A(N).

CALL SLESUM (S, N, A, Y)

The sum will be stored in Y.

B.2 Argument Definitions

X [in] Argument of the polynomials.

N [in] Highest degree of polynomials in sum.

A() [in] The coefficients must be given in A(J), J = 0, ..., N.

Y [out] Computed value of the sum.

B.3 Modifications for Double Precision

For double precision usage, change the REAL statement to DOUBLE PRECISION and change the subroutine name from SLESUM to DLESUM.

C. Examples and Remarks

See DRSLESUM and ODSLESUM for an example of the usage of SLESUM. DRSLESUM evaluates the following identity, the coefficients of which were obtained from Table 22.9, page 798, of [1].

$$z = y - w = 0,$$

where

$$y = 0.07P_0(x) + 0.27P_1(x) + 0.20P_2(x) + 0.28P_3(x) + 0.08P_4(x) + 0.08P_5(x),$$

and

$$w = 0.35x^4 + 0.63x^5.$$

D. Functional Description

The sum is evaluated by the following algorithm:

$$b_{N+2} = 0, \quad b_{N+1} = 0,$$
$$b_k = \frac{2k+1}{k+1} b_{k+1} x - \frac{k+1}{k+2} b_{k+2} + a_k, \quad k = N, ..., 0,$$
$$y = b_0.$$

For an error analysis applying to this algorithm see [2] and [3]. The first four Legendre polynomials are

$$P_0(x) = 1, \quad P_1(x) = x,$$
$$P_2(x) = 1.5x^2 - 0.5, \quad P_3(x) = 2.5x^3 - 1.5x.$$

For $k \geq 2$ the Legendre polynomials satisfy the recurrence

$$kP_k(x) = (2k-1)xP_{k-1}(x) - (k-1)P_{k-2}(x).$$

The Legendre polynomials are orthogonal relative to integration over the interval $[-1, 1]$ and are normally used only with an argument, x, in this interval.

References

1. Milton Abramowitz and Irene A. Stegun, **Handbook of Mathematical Functions**, *Applied Mathematics Series 55*, National Bureau of Standards (1966) Chapter 22, 771–802.

2. E. W. Ng, *Direct summation of series involving higher transcendental functions*, **J. Comp. Phys. 3**, 2 (Oct. 1968) 334–338.

3. E. W. Ng, *Recursive algorithm for the computation of hypergeometric series*, **SIAM J. on Math. Anal. 2** (1971) 31–36.

E. Error Procedures and Restrictions

The subroutine will return Y = 0 if N < 0. It is recommended that x satisfy $|x| \leq 1$.

F. Supporting Information

The source language is ANSI Fortran

Entry	Required Files
DLESUM	DLESUM
SLESUM	SLESUM

Based on a 1974 program by E. W. Ng, JPL. Present version by C. L. Lawson and S. Y. Chiu, JPL, 1983.

DRSLESUM

```
c      DRSLESUM
c>> 1995-05-28 DRSLESUM Krogh  Changes to use M77CON
c>> 1994-08-09 DRSLESUM WVS  Set up for CHGTYP
c>> 1994-07-14 DRSLESUM CLL
c>> 1992-04-29 DRSLESUM CAO Replaced '1' in format.
c>> 1991-11-19 DRSLESUM CLL
c>> 1987-12-09 DRSLESUM Lawson  Initial Code.
c--S replaces "?": ?LESUM, DR?LESUM
c
c     Demonstration driver for evaluation of a Legendre series.
c     ------------------------------------------------------------------
      integer j
      real            x,a(0:5),y,w,z
      data a/0.07e0, 0.27e0, 0.20e0, 0.28e0, 0.08e0, 0.08e0/
c     ------------------------------------------------------------------
      print '(1x,3x,a1,14x,a1,17x,a1/)','x','y','z'
      do 20 j = -10,10,2
        x = real(j) /10.e0
        call slesum (x,5,a,y)
        w = 0.35e0 * (x**4) + 0.63e0 * (x**5)
        z = y - w
        print '(1x,f5.2,5x,g15.7,g15.2)',x,y,z
   20 continue
      end
```

ODSLESUM

```
   x              y                  z

 -1.00     -0.2800000           0.0
 -0.80     -0.6307840E-01        0.22E-07
 -0.60     -0.3628805E-02        0.37E-08
 -0.40      0.2508797E-02       -0.33E-08
 -0.20      0.3583953E-03       -0.48E-08
  0.00       0.000000           0.0
  0.20      0.7616058E-03        0.57E-08
  0.40      0.1541121E-01        0.11E-07
  0.60      0.9434883E-01        0.22E-07
  0.80      0.3497985            0.30E-07
  1.00      0.9800001            0.12E-06
```